

Bild 5.1 : Makromodell eines bipolaren Operationsverstärkers

Datenblattbezeichnung	Formelz./Einh.	uA741
Kompensationskapazität	C_2 / F	30 pF
Slew Rate (pos. Flanke)	$S_R^+ / (V/us)$	0,5 V/us
Slew Rate (neg. Flanke)	$S_R^- / (V/us)$	0,4 V/us
Eingangsstrom	I_B / A	80 nA
Eingangs-Offset-Strom	I_{BOS} / A	10 nA
Eingangs-Offset-Spannung	V_{OS} / V	1 mV
0 dB-Frequenz	f_{0dB} / Hz	1 MHz
Phasenänd. durch 2. Pol bei f_{0dB}	$\Delta\phi / ^\circ$	5°
Verlustleistung	P_d / W	50 mW
Gleichtaktunterdrückung	CMRR	31095,78(90dB)
Ausgangswid. bei höheren Frequenzen	R_{O-ac} / Ohm	37,5 Ohm
Ausgangswid. bei Gleichspannung	R_{out} / Ohm	75 Ohm
Leerlaufgleichspannungsverstärkung	a_{VD}	200000(106dB)
Kurzschlußstrom am Ausgang	I_{sc} / A	25 mA
max. pos. Ausgangsspannung	V_{out}^+ / V	14 V
max. neg. Ausgangsspannung	V_{out}^- / V	-14 V

Tab. 5.1 : Datenblattangaben zur Berechnung der Makromodell-Parameter

$$\begin{aligned}
 V_T &= 25,85 \text{ mV} \\
 I_{SD1} &= I_{SD3} = I_{SD4} = 8 \cdot 10^{-16} \text{ A} = I_S \text{ in } Q_1 \\
 R_2 &= 100 \text{ kOhm}
 \end{aligned}
 \left. \vphantom{\begin{aligned} V_T \\ I_{SD1} \\ R_2 \end{aligned}} \right\} \text{ vorgegeben}$$

$$\begin{aligned}
 I_{C1} &= I_{C2} = (C_2/2) S_{R^+} = 7,5 \text{ uA} \\
 C_E &= 2I_{C1}/S_{R^-} - C_2 = 7,5 \text{ pF} \\
 I_{B1} &= I_B + I_{BOS}/2 = 85 \text{ nA} \\
 I_{B2} &= I_B - I_{BOS}/2 = 75 \text{ nA} \\
 \beta_1 &= I_{C1}/I_{B1} = 88,235 = \text{BF in } Q_1 \\
 \beta_2 &= I_{C2}/I_{B2} = 100 = \text{BF in } Q_2 \\
 I_{EE} &= I_{B1} + I_{B2} + 2I_{C1} = 15,16 \text{ uA} \\
 R_E &= 200 \text{ V}/I_{EE} = 13,2 \text{ MOhm} \\
 I_{S2} &= I_{S1} e^{V_{OS}/V_T} = 8,3155 \cdot 10^{-16} \text{ A} = I_S \text{ in } Q_2 \\
 1/g_{m1} &= V_T/I_{C1} = 3447 \text{ Ohm} \\
 R_{C1} &= R_{C2} = 1/(2\pi f_{0dB} C_2) = 5305 \text{ Ohm} \\
 R_{e1} &= (R_{C1} - 1/g_{m1})\beta_1/(\beta_1 + 1) = 1837,2 \text{ Ohm} \\
 R_{e2} &= (R_{C1} - 1/g_{m1})\beta_2/(\beta_2 + 1) = 1839,6 \text{ Ohm} \\
 C_1 &= (C_2/2) \tan(\Delta\phi) = 1,3 \text{ pF} \\
 R_p &= (V_{CC} - V_{EE})^2 / (P_d - 2I_{C1}V_{CC} + I_{EE}V_{EE}) = 18,16 \text{ kOhm} \\
 G_a &= 1/R_{C1} = 188,5 \text{ uS} \\
 G_{cm} &= 1/(2R_{C1} \text{ CMRR}) = 3,03 \text{ nS} \\
 R_{o1} &= R_{o-ac} = 37,5 \text{ Ohm} \\
 R_{o2} &= R_{out} - R_{o-ac} = 37,5 \text{ Ohm} \\
 G_b &= a_{VD} R_{C1}/(R_2 R_{o2}) = 282,93 \text{ S} \\
 I_x &= 2I_{C1}G_b R_2 - I_{sc} = 424,155 \text{ A} \\
 I_{SD1} &= I_{SD2} = I_x e^{-R_{o1} I_{sc}/V_T} = 7,5335 \cdot 10^{-14} \text{ A} \\
 V_{CD} &= V_{CC} - V_{out}^+ + V_T \ln(I_{sc}/I_{SD3}) = 1,8032 \text{ V} \\
 V_{ED} &= -V_{EE} + V_{out}^- + V_T \ln(I_{sc}/I_{SD4}) = 1,8032 \text{ V}
 \end{aligned}$$

Tab. 5.2 : Gleichungen zur Berechnung der OP-Makromodellparameter

```

.SUBCKT UA741 2 3 6
* 2 = - E, 3 = + E, 6 = A
VEE 4 0 DC -15
VCC 7 0 DC 15
RC1 7 10 5305
RC2 7 11 5305
RE1 12 14 1837.2
RE2 13 14 1839.6
RE 14 0 13.2MEG
RF 7 4 18.16K
R2 15 0 100K
RO2 16 0 37.5
RO1 16 6 37.5
C2 15 16 30P
C1 10 11 1.3P
CE 14 0 7.5P
D1 16 17 D1
D2 17 16 D1
D3 6 18 D3
D4 19 6 D3
Q1 10 2 12 Q1
Q2 11 3 13 Q2
VCD 7 18 DC 1.8032
VED 19 4 DC 1.8032
IE 14 4 DC 15.16U
GCM1 15 0 2 0 3.03N
GCM2 15 0 3 0 3.03N
GA 15 0 10 11 188.5U
GB 16 0 15 0 282.93
EC 17 0 6 0 1
.MODEL D1 D IS=7.5335E-14
.MODEL D3 D IS=8E-16
.MODEL Q1 NPN IS=8E-16 BF=88.235
.MODEL Q2 NPN IS=8.3155E-16 BF=100
.ENDS

```

Bild 5.2 : SPICE-Anweisungen zur Beschreibung des Makromodells als Teilschaltung

```

.SUBCKT UA741 2 3 6
* 2=-E , 3=+E , 6=A
VEE 4 0 DC -15
VCC 7 0 DC 15
R1 1 4 1K
R2 15 4 50K
R3 5 4 1K
R4 17 4 5K
R5 18 16 39K
R6 22 23 4.5K
R7 20 23 7.5K
R8 21 4 50K
R9 19 4 50
R10 24 6 25
R11 6 25 50
C 22 14 30P
Q1 9 3 10 NPN
Q2 12 13 10 PNP
Q3 9 2 11 NPN
Q4 14 13 11 PNP
Q5 12 15 1 NPN
Q6 14 15 5 NPN
Q7 7 12 15 NPN
Q8 9 9 7 PNP
Q9 13 9 7 PNP
Q10 13 16 17 NPN
Q11 16 16 4 NPN
Q12 18 18 7 PNP
Q13 14 19 4 NPN
Q14 20 14 21 NPN
Q15 22 18 7 PNP
Q16 22 23 20 NPN
Q17 20 21 19 NPN
Q18 22 24 6 NPN
Q19 7 22 24 NPN
Q20 4 20 25 PNP
Q21 6 6 23 NPN
.MODEL NPN NPN BF=160 RB=100 CJS=2P
+TF=0.3N TR=6N CJE=3P CJC=2P VAF=100
.MODEL NPN NPN BF=160 RB=100 CJS=2P
+TF=0.3N TR=6N CJE=3P CJC=2P VAF=100 IS=2P
.MODEL PNP PNP BF=20 RB=20 TF=1N TR=20N
+CJE=6P CJC=4P VAF=100
.MODEL PNP PNP BF=20 RB=20 TF=1N TR=20N
+CJE=6P CJC=4P VAF=100 IS=2P
.ENDS

```

Bild 5.4 : SPICE-Anweisungen zur Beschreibung des Device-Modells als Teilschaltung




Bild 5.3 : Schaltbild des Operationsverstärkers uA741 (Device-Modell)

Entnommen aus dem Buch
 E.E.E.Hoefer H.Nielinger
 SPICE
 Analyseprogramm für elektronische Schaltungen
 ISBN 3-540-15160-5 Springer Verlag